

# Risanje fraktalov z Mathematico

Rišemo fraktale, ki so *samopodobni*: celota je sestavljena iz kosov, ki so podobni celoti.

## Podobnostne preslikave

### Razteg

Razteg za faktor  $k$  v izhodišču:

```
ClearAll[raz]
raz[k_, {x_, y_}] := {k * x, k * y}

ClearAll[razteg]
razteg[k_, Point[p_]] := Point[raz[k, p]]
razteg[k_, Circle[p_, r_]] := Circle[raz[k, p], Abs[k * r]]
razteg[k_, Line[lst_]] := Line[Map[Function[p, raz[k, p]], lst]]
razteg[k_, Rectangle[p_, q_]] := Rectangle[raz[k, p], raz[k, q]]
razteg[k_, Graphics[lst_]] := Graphics[Map[Function[u, razteg[k, u]], lst]]
razteg[k_, lst_List] := Map[Function[u, razteg[k, u]], lst]
```

### Premik

Premik za vektor  $v = \{v_1, v_2\}$ :

```
ClearAll[pre]
pre[v_, p_] := p + v

ClearAll[premik]
premik[v_, Point[p_]] := Point[pre[v, p]]
premik[v_, Circle[p_, r_]] := Circle[pre[v, p], r]
premik[v_, Line[lst_]] := Line[Map[Function[p, pre[v, p]], lst]]
premik[v_, Rectangle[p_, q_]] := Rectangle[pre[v, p], pre[v, q]]
premik[v_, Graphics[lst_]] := Graphics[Map[Function[u, premik[v, u]], lst]]
premik[v_, lst_List] := Map[Function[u, premik[v, u]], lst]
```

### Zasuk

Zasuk okoli izhodišča za kot  $\alpha$ :

```
ClearAll[zas]
zas[a_, {x_, y_}] := {Cos[a] * x - Sin[a] * y, Sin[a] * x + Cos[a] * y}

ClearAll[zasuk]
zasuk[a_, Point[p_]] := Point[zas[a, p]]
zasuk[a_, Circle[p_, r_]] := Circle[zas[a, p], r]
zasuk[a_, Line[lst_]] := Line[Map[Function[p, zas[a, p]], lst]]
zasuk[a_, Graphics[lst_]] := Graphics[Map[Function[u, zasuk[a, u]], lst]]
zasuk[a_, lst_List] := Map[Function[u, zasuk[a, u]], lst]
```

## Trikotnik Sierpinskega:


```

ClearAll[sier]
sier[lik_] :=
  {razteg[1/2, lik],
   premik[{0, 1/2}, razteg[1/2, lik]],
   premik[{1/2, 1/2}, razteg[1/2, lik]]
  } // N

ClearAll[sierpinski]
sierpinski[n_] := Graphics[Nest[sier, {Line[{{0, 0}, {1, 0}, {1, 1}, {0, 1}, {0, 0}]}], n]]

sierpinski[7] // Show

```



## Kochova snezinka

Kochova daljica:


```

ClearAll[kdal]
kdal[lst_] :=
  {razteg[1/3, lst],
   premik[{1/3, 0}, razteg[1/3, zasuk[Pi/3, lst]]],
   premik[{1/2, Sqrt[3]/6}, razteg[1/3, zasuk[-Pi/3, lst]]],
   premik[{2/3, 0}, razteg[1/3, lst]]} // N

ClearAll[kochDaljica]
kochDaljica[n_] := Graphics[Nest[kdal, {Line[{{0, 0}, {1, 0}]}], n]]

kochDaljica[5] // Show

```



---

## Naloge

### Naloga: zrcaljenje

V razdelek "Podobnostne preslikave" dodaj še preslikavo **zrcali**  $[ \{ \mathbf{a}, \mathbf{b} \}, \mathbf{l} ]$ , ki prezrcali lik **l** čez premico s smernikom  $\{a, b\}$ .

### Naloga: Kochova snežinka

Sestavi funkcijo **kochSnezinka**, ki nariše Kochovo snežinko.

### Naloga: Preproga Sierpinskega

Sestavi funkcijo, ki nariše preprogo Sierpinskega, glej [http://en.wikipedia.org/wiki/Sierpinski\\_carpet](http://en.wikipedia.org/wiki/Sierpinski_carpet).

### Naloga: Havajski uhan

Sestavi funkcijo, ki nariše Havajski uhan, glej [http://en.wikipedia.org/wiki/Hawaiian\\_earring](http://en.wikipedia.org/wiki/Hawaiian_earring).

### Naloga: ostali fraktali

Sestavi funkcije, ki rišejo razne fraktale. Izberi kak fraktal s seznama [http://en.wikipedia.org/wiki/List\\_of\\_fractals\\_by\\_Hausdorff\\_dimension](http://en.wikipedia.org/wiki/List_of_fractals_by_Hausdorff_dimension), na primer [http://en.wikipedia.org/wiki/Apollonian\\_gasket](http://en.wikipedia.org/wiki/Apollonian_gasket)